



Database 2



Nested Tables الجداول المتداخلة



تعريف المؤشر في ترويسة FOR

```
DECLARE
  CURSOR cur IS (SELECT ename, sal FROM emp WHERE deptno = 10);
  rec_var cur%ROWTYPE;
BEGIN
  FOR rec_var IN cur LOOP
    DBMS_OUTPUT.PUT_LINE(rec_var.ename);
  END LOOP;
END;
/
```

```
BEGIN
  FOR rec_var IN
    (SELECT ename, sal FROM emp WHERE deptno = 10)
  LOOP
    DBMS_OUTPUT.PUT_LINE(rec_var.ename);
  END LOOP;
END;
/
```

FETCH/INTO مع عدة متحولات

```
DECLARE
  CURSOR emp_cur IS
    SELECT ename, sal FROM emp;
  emp_rec emp_cur%ROWTYPE;
BEGIN
  OPEN emp_cur;
  LOOP
    FETCH emp_cur INTO emp_rec;
    EXIT WHEN emp_cur%NOTFOUND;
    DBMS_OUTPUT.PUT_LINE(emp_rec.ename || ', ' || emp_rec.sal);
  END LOOP;
  CLOSE emp_cur;
END;
/
```

FETCH/INTO مع عدة متحولات

```
DECLARE
  CURSOR emp_cur IS
    SELECT ename, sal FROM emp;
  v_ename emp.ename%TYPE;
  v_sal   emp.sal%TYPE;
BEGIN
  OPEN emp_cur;
  LOOP
    FETCH emp_cur INTO v_ename, v_sal;
    EXIT WHEN emp_cur%NOTFOUND;
    DBMS_OUTPUT.PUT_LINE(v_ename||', '||v_sal);
  END LOOP;
  CLOSE emp_cur;
END;
/
```

المؤشرات بوسطاء

```
DECLARE
    CURSOR emp_cur (dno NUMBER) IS
        SELECT ename, sal FROM emp WHERE deptno = dno;
    emp_rec emp_cur%ROWTYPE;
    d dept.deptno%TYPE;
BEGIN
    SELECT deptno INTO d
    FROM dept WHERE DNAME = 'SALES';
    OPEN emp_cur (d);
    LOOP
        FETCH emp_cur INTO emp_rec;
        EXIT WHEN emp_cur%NOTFOUND;
        DBMS_OUTPUT.PUT_LINE (emp_rec.ename || ', ' || emp_rec.sal);
    END LOOP;
    CLOSE emp_cur;
END;
/
```

قفل السجلات باستخدام FOR UPDATE

- يمكن استخدام المؤشرات لإحضار سجلات من أجل التعديل أو الحذف
- تسمح تعليمة **FOR UPDATE** بوضع قفل على سجلات المؤشر
- لا يمكن لمستخدم آخر تعديل هذا السجل قبل أن يتم حفظه بشكل دائم و إزالة القفل
- من خلال إغلاق المؤشر
- أو باستخدام تعليمة التثبيت **COMMIT** أو التراجع **ROLLBACK**

```
CURSOR cur IS <select_statment> FOR UPDATE;
```

قفل السجلات باستخدام FOR UPDATE

```
DECLARE
  CURSOR a IS
    SELECT ename FROM emp WHERE deptno = 10
    FOR UPDATE;
  a_var a%ROWTYPE;
BEGIN
  OPEN a;
  FETCH a INTO a_var;
  WHILE a%FOUND LOOP
    DBMS_OUTPUT.PUT_LINE(a_var.ename);
    FETCH a INTO a_var;
  END LOOP;
  CLOSE a;
END;
/
```

تعديل السجلات باستخدام CURRENT OF

- عند إحضار السجلات من مؤشر يتم وضع علامة لتحديد السجل الحالي
- إذا أردنا تعديل هذا السجل، يمكن استخدام CURRENT OF بعد تعليمة UPDATE أو DELETE

```
UPDATE <table_name> SET <column_name> = <value>  
WHERE CURRENT OF <cursor_name>;
```


تعديل السجلات باستخدام CURRENT OF

```
DECLARE
  CURSOR a IS
    SELECT ename, sal FROM emp WHERE deptno = 10
    FOR UPDATE;
  a_var a%ROWTYPE;
BEGIN
  OPEN a;
  FETCH a INTO a_var;
  WHILE a%FOUND LOOP
    DBMS_OUTPUT.PUT_LINE(a_var.ename||' '||a_var.sal);
    UPDATE emp SET sal = sal * 1.03
    WHERE CURRENT OF a;
    FETCH a INTO a_var;
  END LOOP;
END;
/
```

الجداول المتداخلة

Nested Tables

تعريف جدول متداخل

```
CREATE OR REPLACE TYPE Address_TY AS OBJECT (  
    Street    VARCHAR2(50) ,  
    City      VARCHAR2(30) ,  
    State     VARCHAR2(2) ,  
    Zip_code  VARCHAR2(10)  
)  
/  
  
CREATE TYPE Address_NT AS TABLE OF Address_TY  
/  
  
CREATE TABLE Person (  
    name      VARCHAR2(25) ,  
    address   Address_NT  
)  
NESTED TABLE address STORE AS address_nt_tab  
;
```

تعريف جدول متداخل

Person				
name	address			
	Address_NT_TAB			
	Street	City	State	Zip_Code

إدراج أسطر في جدول متداخل

```
INSERT INTO Person VALUES
('John Smith',
  Address_NT(
    Address_TY('2145 W. Oak', 'Seattle', 'WA', '98042')
  )
);
```

```
INSERT INTO Person VALUES
('Robert Freeman',
  Address_NT(
    Address_TY('550 Water St.', 'Jax', 'FL', '32202'),
    Address_TY('37 Black St.', 'Detroit', 'NY', '67324')
  )
);
```

الاستعلام على الجداول المتداخلة

```
SELECT Street  
FROM Address_NT  
WHERE City = 'Seattle';
```

```
FROM Address_NT
```

```
*
```

```
ERROR at line 2:
```

```
ORA-04044: procedure, function, package, or type is not allowed here
```

```
SELECT Street  
FROM Address_NT_TAB  
WHERE City = 'Seattle';
```

```
FROM Address_NT_TAB
```

```
*
```

```
ERROR at line 2:
```

```
ORA-22812: cannot reference nested table column's storage table
```

الاستعلام على الجداول المتداخلة

```
SELECT NT.Street
FROM THE (SELECT address
          FROM Person
          WHERE name = 'John Smith') NT
WHERE NT.City = 'Seattle';
```

Street
2145 W. Oak

استخدامات أخرى للتابع THE

إدخال سطر جديد في جدول العناوين الخاص بشخص سبق إدخاله

```
INSERT INTO
THE (SELECT address
     FROM Person
     WHERE name = 'John Smith')
VALUES
(Address_TY('21 Writer Sq.', 'New York', 'NY', '65324')) ;
```


الإدراج باستخدام الاستفسارات

مثال : إدراج شخص جديد له نفس عناوين شخص آخر سبق إدخاله

```
INSERT INTO Person VALUES
('Mary Smith',
  CAST (
    MULTISET (
      SELECT *
      FROM THE (SELECT Address
                FROM Person
                WHERE name = 'John Smith')
      ) AS Address_NT
    )
);
```

- **CAST** : تسمح بقولبة نتائج الاستفسار ليصبح كشكل جدول
- **MULTISET** : تسمح للاستفسار المقولب أن يحوي عدة سجلات

الإدراج باستخدام الاستفسارات

```
SELECT *  
FROM THE ( SELECT address  
            FROM Person  
            WHERE Name = 'Mary Smith' );
```

City	Street	State	Zip_Code
Seattle	2145 W. Oak	WA	٩٨٠٤٢
New York	21 Writer Sq.	NY	٦٥٣٢٤